
    Spector

CQRS-style event sourcing for Ecto schemas.
Spector records all changes to your Ecto schemas as events in a separate event log table. This enables full audit trails, temporal queries, and the ability to replay history. For tamper-evident logs, enable optional hash chain integrity.
Installation
Add spector to your list of dependencies in mix.exs:
def deps do
  [
    {:spector, "~> 0.8.0"}
  ]
end
Quick Start
1. Define an Events Table
defmodule MyApp.Events do
  use Spector.Events,
    table: "events",
    schemas: [MyApp.User, MyApp.Post],
    repo: MyApp.Repo
end
2. Mark Schemas as Evented
defmodule MyApp.User do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "users" do
    field :name, :string
    field :email, :string
    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Ecto.Changeset.cast(attrs, [:name, :email, :inserted_at, :updated_at])
    |> Ecto.Changeset.validate_required([:name, :email])
  end
end
3. Create Migrations
# For the events table
defmodule MyApp.Repo.Migrations.CreateEvents do
  use Ecto.Migration

  def up, do: Spector.Migration.up(table: "events")
  def down, do: Spector.Migration.down(table: "events")
end
4. Use Spector Instead of Repo
# Insert
{:ok, user} = Spector.insert(MyApp.User, %{name: "Alice", email: "alice@example.com"})

# Update
{:ok, user} = Spector.update(user, %{name: "Alice Smith"})

# Delete
{:ok, user} = Spector.delete(user)
How It Works
When you update or execute an action on a record, Spector "rolls forward" by replaying all stored events through your schema's changeset/2 function. This means:
	Your changeset function handles both new operations AND historical replay
	Schema migrations happen automatically during replay (using version guards)
	The current state is always reconstructed from the event log
	Stale in-memory records are never a problem

This design lets you evolve your schema over time while maintaining full compatibility with historical events.
Features
Custom Actions
Define domain-specific actions beyond insert/update/delete:
defmodule MyApp.Item do
  use Spector.Evented, events: MyApp.Events, actions: [:archive]
  use Ecto.Schema

  schema "items" do
    field :name, :string
    field :value, :integer
    field :archived_at, :utc_datetime_usec
    timestamps()
  end

  def changeset(changeset, attrs) when changeset.action == :archive do
    Ecto.Changeset.change(changeset, archived_at: attrs[:archived_at])
  end

  def changeset(changeset, attrs) do
    Ecto.Changeset.cast(changeset, attrs, [:name, :value, :inserted_at, :updated_at])
  end
end

# Execute custom action
{:ok, item} = Spector.execute(item, :archive, %{archived_at: DateTime.utc_now()})
Schema Versioning
Handle schema migrations with version guards:
defmodule MyApp.User do
  use Spector.Evented, events: MyApp.Events, version: 1
  use Ecto.Schema

  schema "users" do
    field :name, :string
    timestamps()
  end

  # Migrate v0 events (with :title) to v1 (with :name)
  def changeset(changeset, attrs) when version_is(attrs, 0) do
    attrs = Map.put(attrs, "name", attrs["title"])
    do_changeset(changeset, attrs)
  end

  def changeset(changeset, attrs), do: do_changeset(changeset, attrs)

  defp do_changeset(changeset, attrs) do
    Ecto.Changeset.cast(changeset, attrs, [:name, :inserted_at, :updated_at])
  end
end
Savepoints
Capture full record state at specific points for faster replay:
defmodule MyApp.User do
  @behaviour Spector.Evented
  use Spector.Evented, events: MyApp.Events

  @impl true
  def savepoint(record, _version) do
    %{name: record.name, email: record.email}
  end
end

# Create a savepoint
{:ok, user} = Spector.savepoint(MyApp.User, user_id)
When replaying events, Spector starts from the most recent savepoint instead of the beginning, improving performance for records with long histories.
Materializing Records
Recreate a database record from its event history:
# If the record was deleted from the database but events remain
{:ok, user} = Spector.materialize(MyApp.Events, user_id)
This replays all events for the given parent_id and inserts the resulting record into the database. Useful for recovering deleted records or populating a database from an event log.
Preparing Materialization
Use the prepare_materialization/1 callback to modify the final changeset before database insert/update. This is useful for setting associations from accumulated data:
defmodule MyApp.Task do
  @behaviour Spector.Evented
  use Spector.Evented, events: MyApp.Events

  schema "tasks" do
    field :title, :string
    field :assignee_ids, {:array, :binary_id}, virtual: true
    many_to_many :assignees, MyApp.User, join_through: "task_assignees"
  end

  @impl true
  def prepare_materialization(changeset) do
    assignee_ids = Ecto.Changeset.get_field(changeset, :assignee_ids) || []
    users = Enum.map(assignee_ids, &%MyApp.User{id: &1})
    Ecto.Changeset.put_assoc(changeset, :assignees, users)
  end
end
This callback is primarily for database-backed schemas. For embedded schemas, it is not called automatically but may be defined for manual use.
Hash Chain Integrity
Enable tamper-evident event logs with cryptographic hashing:
defmodule MyApp.Events do
  use Spector.Events,
    table: "events",
    schemas: [MyApp.User],
    repo: MyApp.Repo,
    hashed: true
end
Integrity Verification
Verify the integrity of your event logs:
# Verify all savepoints for a record
:ok = Spector.Integrity.verify_savepoints(MyApp.User, user_id)

# Verify hash chain for entire events table
:ok = Spector.Integrity.verify_hash_chain(MyApp.Events)
Explicit Schema Indexing
Ensure stability when adding/removing schemas:
schemas: [MyApp.User, MyApp.Post, {MyApp.Comment, 10}]
Action Aliases
Maintain backwards compatibility when renaming actions:
use Spector.Events,
  aliases: [soft_delete: :archive]
Chat and Conversation Logs
Spector includes features specifically designed for chat-log style applications:
	Event links for tracking message ancestry and edit history
	Embedded schemas for state reconstructed purely from events

See the AI Chat Guide for conversation branching and the Basic Chat Guide for edit history tracking.
Guides
	Building an AI Chat Log - Conversation branching with tree-structured message history
	Associations Guide - Handling Ecto associations with event-sourced schemas
	Building a Basic Chat - Simple chat with edit history tracking
	Event Links Guide - Creating many-to-many relationships between events
	Otzel Integration - Rich text editing with Operational Transformation

Database Support
Spector works with any database supported by Ecto for basic functionality.
Note: Hashed event tables (hashed: true) currently require PostgreSQL. The hash chain integrity feature uses LOCK TABLE ... IN EXCLUSIVE MODE which is PostgreSQL-specific.
Documentation
Full documentation is available at HexDocs.
License
MIT License. See LICENSE for details.


  

    Building an AI Chat Log with Spector

This guide shows how to use Spector to build an AI chat log that supports conversation branching (like ChatGPT's "edit and regenerate" feature).
Overview
AI chat applications often need:
	Full conversation history
	Ability to branch from any point in the conversation
	Reconstruct any conversation path from the event log

Spector's event sourcing with links provides exactly this. Each message becomes an event, and branching is handled by tracking ancestors.
Setup
1. Define the Chat Schema
Since chat state is reconstructed from events, use an embedded schema (no database table):
defmodule MyApp.AIChat do
  use Spector.Evented, events: MyApp.AIChatEvents, actions: [:append]
  use Ecto.Schema
  alias Ecto.Changeset

  @behaviour Spector.Evented

  defmodule Message do
    use Ecto.Schema

    @primary_key false
    embedded_schema do
      field :id, :binary_id
      field :content, :string
      field :role, Ecto.Enum, values: [:user, :assistant, :system]
    end

    def changeset(message \\ %__MODULE__{}, attrs) do

      message
      |> Changeset.cast(attrs, [:id, :content, :role])
      |> Spector.changeset_put_event_id(attrs)
      |> Changeset.validate_required([:id, :content, :role])
    end
  end

  @primary_key {:id, :binary_id, autogenerate: false}
  embedded_schema do
    embeds_many :messages, Message
  end

  @impl true
  def prepare_event(event_changeset, previous_events, %{to: ancestor_id}) do
    # Find the ancestor event and build the ancestry chain
    previous_events
    |> Enum.find(&(&1.id == ancestor_id))
    |> MyApp.Repo.preload(:ancestors)
    |> then(&[&1 | &1.ancestors])
    |> then(&Changeset.put_assoc(event_changeset, :ancestors, &1))
  end

  def prepare_event(event_changeset, _previous_events, _attrs) do
    # Require :to for append actions (except the first message)
    if Changeset.fetch_field!(event_changeset, :action) == :append do
      raise ArgumentError, "Missing :to attribute for append action"
    else
      event_changeset
    end
  end

  def changeset(changeset \\ %__MODULE__{}, attrs) do
    changeset = Changeset.change(changeset)

    message =
      attrs
      |> Message.changeset()
      |> Changeset.apply_action!(:insert)

    current_messages = Changeset.get_field(changeset, :messages, [])
    Changeset.put_change(changeset, :messages, [message | current_messages])
  end

  def get_branch(tail_id) do
    case get_branch_events(tail_id) do
      [%{parent_id: parent_id} | _] = events ->
        events
        |> Enum.reduce(%__MODULE__{id: parent_id}, fn event, acc ->
          changeset(acc, Map.put(event.payload, "__event_id__", event.id))
        end)
        |> Changeset.apply_action!(:get)
      [] -> nil
    end
  end

  defp get_branch_events(tail_id) do
    import Ecto.Query

    ancestor_ids =
      from(link in "ai_chat_ancestors",
        where: link.event_id == type(^tail_id, :binary_id),
        select: link.ancestor_id
      )

    MyApp.Repo.all(
      from(e in MyApp.AIChatEvents,
        where: e.id == ^tail_id or e.id in subquery(ancestor_ids),
        order_by: [desc: e.inserted_at]
      )
    )
  end
end
2. Define the Events Module
defmodule MyApp.AIChatEvents do
  use Spector.Events,
    table: "ai_chat_events",
    links: [ancestors: {"ai_chat_ancestors", :ancestor_id}],
    schemas: [MyApp.AIChat],
    repo: MyApp.Repo
end
The links option creates a many-to-many relationship for tracking message ancestry. Each event can reference its ancestor events, enabling tree-structured conversations.
3. Create the Migration
defmodule MyApp.Repo.Migrations.CreateChatEvents do
  use Ecto.Migration

  def up do
    Spector.Migration.up(
      table: "ai_chat_events",
      links: [{"ai_chat_ancestors", :ancestor_id}]
    )
  end

  def down do
    Spector.Migration.down(
      table: "ai_chat_events",
      links: [{"ai_chat_ancestors", :ancestor_id}]
    )
  end
end
Usage
Starting a Conversation
assert {:ok, %MyApp.AIChat{messages: [%{role: :user, content: "Hello, how are you?"}]} = chat} =
  Spector.insert(MyApp.AIChat, %{
    role: :user,
    content: "Hello, how are you?"
  })
Adding Messages
To add a message, use Spector.execute/3 with the :append action. The :to attribute specifies which message to append after:
# Append assistant response
assert {:ok, %{messages: [%{role: :assistant, content: "I'm doing well, thank you!"}, _]} = chat} =
  Spector.execute(chat, :append, %{
    to: chat.id,
    role: :assistant,
    content: "I'm doing well, thank you!"
  })
Branching a Conversation
To branch from an earlier point, specify the :to attribute pointing to the message you want to branch from:
# branch from the first message
{:ok, %{messages: [%{id: branch_id}, %{id: base_id}, _]}} = Spector.execute(chat, :append, %{
  to: chat.id,
  role: :user,
  content: "Actually, let me rephrase that..."
})
Reconstructing a Conversation Branch
Use get_branch/1 (defined in the Chat module above) to retrieve the chat state at a specific branch point. Note that messages are prepended in our implementation above, so the most recent message appears first:
# Get the assistant's branch (original user message + assistant response)
assert %{messages: [
  %{role: :user, content: "Hello, how are you?"}, 
  %{role: :assistant, content: "I'm doing well, thank you!"}]} = MyApp.AIChat.get_branch(base_id)

# Get the new branch (original user message + rephrased user message)
assert %{messages: [
  %{role: :user, content: "Hello, how are you?"}, 
  %{role: :user, content: "Actually, let me rephrase that..."}]} = MyApp.AIChat.get_branch(branch_id)
How It Works
	Insert: Creates the first event with the initial message. The event id and parent_id are the same (this is the conversation root).

	Append: Creates a new event linked to the conversation (parent_id) with ancestry tracking to the specified message (to).

	Branching: When you append with a :to pointing to an earlier message, prepare_event/3 builds the ancestry chain from that point. This creates a new branch in the conversation tree.

	Reconstruction: The ancestors association lets you traverse back through any branch to reconstruct the full conversation path.


Tips
	The role enum (:user, :assistant, :system) matches common AI API conventions
	Consider adding timestamps to message payloads for display purposes
	The parent_id groups all events for a conversation; use it to list all branches
	Use get_branch/1 for reconstructing a single path through the conversation tree



  

    Associations Guide

This guide covers how to handle Ecto associations with Spector event-sourced schemas. The key insight is that event sourcing stores changes to individual records, so associations work differently than in traditional Ecto applications.
Overview
Event sourcing captures changes as immutable events. When dealing with associations, you need to consider:
	Where is the foreign key stored? The schema that stores the foreign key is straightforward to event-source.
	Can the relationship be rebuilt from events? If replaying events can't reconstruct the relationship, you need a different approach.
	Who owns the relationship data? This determines which schema's events capture changes.

belongs_to (Straightforward)
belongs_to associations are the simplest case because the foreign key is stored directly in the schema that declares the association.
Schema Definition
defmodule MyApp.Item do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "items" do
    field :title, :string
    belongs_to :workspace, MyApp.Workspace
    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :workspace_id])
    |> Changeset.validate_required([:title, :workspace_id])
    |> Changeset.foreign_key_constraint(:workspace_id)
  end
end
Usage
# Create an item belonging to a workspace
{:ok, item} = Spector.insert(MyApp.Item, %{
  title: "New Task",
  workspace_id: workspace.id
})

# The workspace_id is stored in the event payload
# Replaying events correctly rebuilds the association
The foreign key (workspace_id) is cast like any other field. Events capture changes to this field, and replaying events correctly reconstructs which workspace the item belongs to.
Why Not has_one
has_one associations are problematic with event sourcing and are generally not recommended.
The Problem
With has_one, the "owned" side stores the foreign key, not the owner:
defmodule MyApp.User do
  has_one :profile, MyApp.Profile  # Profile stores user_id
end

defmodule MyApp.Profile do
  belongs_to :user, MyApp.User  # user_id is here
end
When you replay events for a User, you only see changes to the User schema. The user_id field lives in Profile, so the User's event history knows nothing about whether a Profile exists or which one it points to.
The Same Problem as has_many
has_one has the same conceptual issues as has_many - the "many" side stores the foreign key. The only difference is cardinality, which doesn't help with event replay.
Recommendations
	Use belongs_to from the other side: Query the Profile by user_id when you need it
	Embed the data: If the profile is truly owned by the user, consider using an embedded schema or JSON field
	Accept the limitation: Preload the association separately after rebuilding from events

has_many (Child Stores Parent ID)
has_many associations work well when each child is event-sourced independently.
The Pattern
defmodule MyApp.Item do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "items" do
    field :title, :string
    has_many :comments, MyApp.Comment
    timestamps()
  end
end

defmodule MyApp.Comment do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "comments" do
    field :body, :string
    belongs_to :item, MyApp.Item
    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:body, :item_id])
    |> Changeset.validate_required([:body, :item_id])
  end
end
Why This Works
	Each Comment is an independent event-sourced entity
	Comments store their own item_id
	Replaying Comment events rebuilds which Item each Comment belongs to
	No special Spector handling needed

Usage
# Create a comment on an item
{:ok, comment} = Spector.insert(MyApp.Comment, %{
  body: "Great work!",
  item_id: item.id
})

# Query comments for an item
comments = MyApp.Repo.all(
  from c in MyApp.Comment,
    where: c.item_id == ^item.id
)

# Or preload through the association
item = MyApp.Repo.preload(item, :comments)
many_to_many (Virtual Field Pattern)
many_to_many associations require special handling because the relationship data lives in a join table, not in either schema.
The Challenge
Events capture changes to schema fields. A join table is a separate entity, so changes to relationships aren't automatically tracked. The solution is to:
	Store relationship IDs in a virtual field
	Capture changes to this virtual field in events
	Sync to the join table using prepare_materialization/1

Schema Definition
defmodule MyApp.Item do
  @behaviour Spector.Evented
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "items" do
    field :title, :string
    field :member_ids, {:array, :binary_id}, virtual: true, default: []
    many_to_many :members, MyApp.Member, join_through: "item_members"
    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :member_ids])
    |> Changeset.validate_required([:title])
  end

  @impl true
  def prepare_materialization(changeset) do
    member_ids = Changeset.get_field(changeset, :member_ids) || []
    members = Enum.map(member_ids, &%MyApp.Member{id: &1})
    Changeset.put_assoc(changeset, :members, members)
  end
end
Join Table Migration (No Schema)
This simple case uses join_through: "item_members" with a table name string. There is no Ecto schema for the join table—Ecto manages it directly. This works when you only need to track which records are related, with no additional metadata.
defmodule MyApp.Repo.Migrations.CreateItemMembers do
  use Ecto.Migration

  def change do
    create table(:item_members, primary_key: false) do
      add :item_id, references(:items, type: :binary_id, on_delete: :delete_all), null: false
      add :member_id, references(:members, type: :binary_id, on_delete: :delete_all), null: false
    end

    create unique_index(:item_members, [:item_id, :member_id])
    create index(:item_members, [:member_id])
  end
end
How It Works
	On insert/update: The member_ids virtual field is cast into the changeset
	Event creation: The member_ids array is stored in the event payload
	On replay: Events rebuild the member_ids virtual field
	Materialization: prepare_materialization/1 converts IDs to association structs
	Database sync: put_assoc handles the join table insert/update/delete

Usage
# Create with members
{:ok, item} = Spector.insert(MyApp.Item, %{
  title: "Team Task",
  member_ids: [user1.id, user2.id]
})

# Update members
{:ok, item} = Spector.update(%{item | member_ids: [user1.id, user2.id, user3.id]})

# The event log captures:
# 1. Insert event with member_ids: [user1.id, user2.id]
# 2. Update event with member_ids: [user1.id, user2.id, user3.id]
Handling Updates (Preloading)
For updates, you need to preload existing member IDs into the virtual field:
def get_item_for_update(id) do
  item = MyApp.Repo.get!(MyApp.Item, id) |> MyApp.Repo.preload(:members)
  %{item | member_ids: Enum.map(item.members, & &1.id)}
end

# Now updates work correctly
item = get_item_for_update(item.id)
{:ok, updated} = Spector.update(%{item | member_ids: new_member_ids})
Join Table with Join Schema (Metadata on Association)
When you need to store metadata on the relationship itself (e.g., role, permissions, timestamps), use a join schema instead of a bare table name.
Join Schema Definition
defmodule MyApp.ItemMember do
  use Ecto.Schema
  alias Ecto.Changeset

  @primary_key false
  schema "item_members" do
    belongs_to :item, MyApp.Item, primary_key: true
    belongs_to :member, MyApp.Member, primary_key: true
    field :role, :string, default: "viewer"
    field :added_at, :utc_datetime
  end

  def changeset(item_member \\ %__MODULE__{}, attrs) do
    item_member
    |> Changeset.cast(attrs, [:item_id, :member_id, :role, :added_at])
    |> Changeset.validate_required([:item_id, :member_id])
    |> Changeset.validate_inclusion(:role, ["viewer", "editor", "admin"])
  end
end
Updated Item Schema
Instead of storing just IDs, store maps with the metadata:
defmodule MyApp.Item do
  @behaviour Spector.Evented
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "items" do
    field :title, :string
    # Store member data as maps: %{member_id: uuid, role: "editor"}
    field :member_data, {:array, :map}, virtual: true, default: []
    many_to_many :members, MyApp.Member, join_through: MyApp.ItemMember
    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :member_data])
    |> Changeset.validate_required([:title])
  end

  @impl true
  def prepare_materialization(changeset) do
    member_data = Changeset.get_field(changeset, :member_data) || []
    item_id = Changeset.get_field(changeset, :id)

    item_members = Enum.map(member_data, fn data ->
      %MyApp.ItemMember{
        item_id: item_id,
        member_id: data["member_id"],
        role: data["role"] || "viewer",
        added_at: parse_datetime(data["added_at"])
      }
    end)

    Changeset.put_assoc(changeset, :members, item_members)
  end

  defp parse_datetime(nil), do: nil
  defp parse_datetime(str) when is_binary(str), do: DateTime.from_iso8601(str) |> elem(1)
  defp parse_datetime(%DateTime{} = dt), do: dt
end
Usage
# Create with members and roles
{:ok, item} = Spector.insert(MyApp.Item, %{
  title: "Team Project",
  member_data: [
    %{member_id: alice.id, role: "admin", added_at: DateTime.utc_now()},
    %{member_id: bob.id, role: "editor", added_at: DateTime.utc_now()}
  ]
})

# Update a member's role
item = get_item_for_update(item.id)
updated_data = Enum.map(item.member_data, fn
  %{"member_id" => id} = data when id == bob.id ->
    Map.put(data, "role", "admin")
  data ->
    data
end)
{:ok, item} = Spector.update(%{item | member_data: updated_data})
Preloading with Metadata
def get_item_for_update(id) do
  item = MyApp.Repo.get!(MyApp.Item, id)
    |> MyApp.Repo.preload(members: :item_member)

  member_data = Enum.map(item.members, fn member ->
    # Access the join record through the preloaded association
    join = Enum.find(member.item_members, &(&1.item_id == item.id))
    %{
      "member_id" => member.id,
      "role" => join.role,
      "added_at" => join.added_at
    }
  end)

  %{item | member_data: member_data}
end
The key difference from the simple case: events capture the full relationship data (including metadata), not just IDs. This lets you replay events and reconstruct not just which members are associated, but how they're associated.
Polymorphic Relations (Advanced)
For relationships between different entity types, store type+id pairs.
Schema Definition
defmodule MyApp.Item do
  @behaviour Spector.Evented
  use Spector.Evented, events: MyApp.Events, actions: [:add_relation, :remove_relation]
  use Ecto.Schema
  alias Ecto.Changeset

  schema "items" do
    field :title, :string
    # Store relations as maps: %{type: "item", id: uuid}
    field :relations, {:array, :map}, virtual: true, default: []
    timestamps()
  end

  def changeset(changeset, attrs) when changeset.action == :add_relation do
    current = Changeset.get_field(changeset, :relations) || []
    new_relation = %{
      "type" => Spector.get_attr(attrs, :relation_type),
      "id" => Spector.get_attr(attrs, :relation_id)
    }
    Changeset.put_change(changeset, :relations, [new_relation | current])
  end

  def changeset(changeset, attrs) when changeset.action == :remove_relation do
    current = Changeset.get_field(changeset, :relations) || []
    remove_id = Spector.get_attr(attrs, :relation_id)
    filtered = Enum.reject(current, &(&1["id"] == remove_id))
    Changeset.put_change(changeset, :relations, filtered)
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :relations])
    |> Changeset.validate_required([:title])
  end

  @impl true
  def prepare_materialization(changeset) do
    # Sync relations to a join table if needed
    relations = Changeset.get_field(changeset, :relations) || []
    item_id = Changeset.get_field(changeset, :id)

    relation_structs = Enum.map(relations, fn rel ->
      %MyApp.ItemRelation{
        item_id: item_id,
        target_type: rel["type"],
        target_id: rel["id"]
      }
    end)

    Changeset.put_assoc(changeset, :item_relations, relation_structs)
  end
end
Join Table
defmodule MyApp.ItemRelation do
  use Ecto.Schema

  @primary_key false
  schema "item_relations" do
    field :item_id, :binary_id
    field :target_type, :string
    field :target_id, :binary_id
  end
end
Usage
# Add a dependency on another item
{:ok, item} = Spector.execute(item, :add_relation, %{
  relation_type: "item",
  relation_id: other_item.id
})

# Add a dependency on a milestone
{:ok, item} = Spector.execute(item, :add_relation, %{
  relation_type: "milestone",
  relation_id: milestone.id
})
Additional Join Table Patterns
has_many :through
For relationships like Item -> ItemMember -> Member -> MemberRole:
# Option 1: Event-source the middle entity
defmodule MyApp.ItemMember do
  use Spector.Evented, events: MyApp.Events

  schema "item_members" do
    belongs_to :item, MyApp.Item
    belongs_to :member, MyApp.Member
    field :role, :string
  end
end

# Option 2: Virtual field on the source
# Store the full path: [{member_id: x, role: y}, ...]
Self-Referential many_to_many
For items that depend on other items:
defmodule MyApp.Item do
  schema "items" do
    field :dependency_ids, {:array, :binary_id}, virtual: true, default: []
    many_to_many :dependencies, MyApp.Item,
      join_through: "item_dependencies",
      join_keys: [source_id: :id, target_id: :id]
  end
end
The join table needs distinct column names:
create table(:item_dependencies, primary_key: false) do
  add :source_id, references(:items, type: :binary_id), null: false
  add :target_id, references(:items, type: :binary_id), null: false
end

create unique_index(:item_dependencies, [:source_id, :target_id])
Temporal/Versioned Joins
For tracking when relationships were active:
defmodule MyApp.ItemMembership do
  use Spector.Evented, events: MyApp.Events

  schema "item_memberships" do
    belongs_to :item, MyApp.Item
    belongs_to :member, MyApp.Member
    field :joined_at, :utc_datetime
    field :left_at, :utc_datetime
  end
end
Event-sourcing the join table itself provides a complete audit trail of relationship changes.
Best Practices
	Prefer belongs_to: The simplest pattern. Put the foreign key on the side you're event-sourcing.

	Event-source children independently: For has_many, let each child be its own event-sourced entity rather than trying to manage the list from the parent.

	Use virtual fields for many_to_many: Store IDs in a virtual field, sync to the join table in prepare_materialization/1.

	Consider embedded schemas: For truly owned data that doesn't need independent identity, embedded schemas or JSON fields avoid association complexity.

	Be explicit about ownership: Decide which side "owns" the relationship and captures changes in its events.

	Preload before updates: For virtual field patterns, load existing relationship IDs into the virtual field before making changes.

	Document your patterns: Association handling varies by use case. Document the pattern each relationship uses.


See Also
	Building a Basic Chat - Uses belongs_to for conversation membership
	Event Links Guide - For event-to-event relationships within the same record



  

    Building a Basic Chat with Spector

This guide shows how to use Spector to build a simple chat with edit history tracking using event links.
Overview
Basic chat applications need:
	Message history with user identification
	Ability to edit messages while preserving previous versions
	Simple linear conversation flow (no branching)

Spector's event sourcing with links tracks every edit as a linked event chain.
Setup
1. Define the Chat Schema
defmodule MyApp.BasicChat do
  use Spector.Evented, events: MyApp.BasicChatEvents, actions: [:edit]
  use Ecto.Schema
  alias Ecto.Changeset
  import Ecto.Query

  @behaviour Spector.Evented

  defmodule Message do
    use Ecto.Schema

    @primary_key false
    embedded_schema do
      field :id, :binary_id
      field :user_id, :binary_id
      field :content, :string
    end

    def changeset(message \\ %__MODULE__{}, attrs) do
      message
      |> Changeset.cast(attrs, [:user_id, :content])
      |> Spector.changeset_put_event_id(attrs)
      |> Changeset.validate_required([:id, :user_id, :content])
    end
  end

  @primary_key {:id, :binary_id, autogenerate: false}
  embedded_schema do
    embeds_many :messages, Message
  end

  @impl true
  def prepare_event(event_changeset, previous_events, %{edits: previous_id}) do
    previous_events
    |> Enum.find(&(&1.id == previous_id))
    |> then(&Changeset.put_assoc(event_changeset, :edits, [&1]))
  end

  def prepare_event(event_changeset, _previous_events, _attrs) do
    event_changeset
  end

  def changeset(chat \\ %__MODULE__{}, attrs)

  def changeset(chat, %{edits: message_id, content: new_content}) when chat.action == :edit do
    chat = Changeset.change(chat)
    current_messages = Changeset.get_field(chat, :messages, [])

    updated_messages =
      Enum.map(current_messages, fn message ->
        if message.id == message_id do
          %{message | content: new_content}
        else
          message
        end
      end)

    Changeset.put_change(chat, :messages, updated_messages)
  end

  def changeset(chat, attrs) do
    chat = Changeset.change(chat)

    message =
      attrs
      |> Message.changeset()
      |> Changeset.apply_action!(:insert)

    current_messages = Changeset.get_field(chat, :messages, [])
    Changeset.put_change(chat, :messages, [message | current_messages])
  end

  def list_messages(chat_id) do
    edited_ids =
      from(link in "basic_chat_edits", select: link.previous_id)
      |> MyApp.Repo.all()

    Spector.all_events(__MODULE__, chat_id)
    |> Enum.reject(&(&1.id in edited_ids))
  end

  def get_edit_history(message_id) do
    previous_ids =
      from(link in "basic_chat_edits",
        where: link.event_id == type(^message_id, :binary_id),
        select: link.previous_id
      )

    MyApp.Repo.all(
      from(e in MyApp.BasicChatEvents,
        where: e.id == ^message_id or e.id in subquery(previous_ids),
        order_by: [asc: e.inserted_at]
      )
    )
  end
end
2. Define the Events Module
defmodule MyApp.BasicChatEvents do
  use Spector.Events,
    table: "basic_chat_events",
    links: [edits: {"basic_chat_edits", :previous_id}],
    schemas: [MyApp.BasicChat],
    repo: MyApp.Repo
end
The edits link creates a relationship from the edited message to its previous version.
3. Create the Migration
defmodule MyApp.Repo.Migrations.CreateBasicChatEvents do
  use Ecto.Migration

  def up do
    Spector.Migration.up(
      table: "basic_chat_events",
      links: [{"basic_chat_edits", :previous_id}]
    )
  end

  def down do
    Spector.Migration.down(
      table: "basic_chat_events",
      links: [{"basic_chat_edits", :previous_id}]
    )
  end
end
Usage
Starting a Conversation
user_id = Ecto.UUID.generate()

assert {:ok, %MyApp.BasicChat{messages: [%{user_id: ^user_id, content: "Hello everyone!"}]} = chat} =
  Spector.insert(MyApp.BasicChat, %{
    user_id: user_id,
    content: "Hello everyone!"
  })
Adding Messages
Note that messages are prepended, so the most recent message appears first:
other_user_id = Ecto.UUID.generate()

assert {:ok, %{messages: [%{user_id: ^other_user_id, content: "Hi there!"}, %{content: "Hello everyone!"}]} = chat} =
  Spector.update(chat, %{
    user_id: other_user_id,
    content: "Hi there!"
  })
Editing a Message
Use the :edit action with :edits pointing to the message ID being edited:
[_, %{id: first_message_id}] = chat.messages

assert {:ok, %{messages: [%{content: "Hi there!"}, %{content: "Hello everyone! (edited)"}]}} =
  Spector.execute(chat, :edit, %{
    edits: first_message_id,
    content: "Hello everyone! (edited)"
  })
How It Works
	Insert: Creates the first event with the initial message.

	Update: Adds new messages to the conversation.

	Edit: Creates a new event linked to the previous version via basic_chat_edits. The original remains in the event log.

	History: Follow the edits links backward to see all versions of a message.


Tips
	The user_id in each message tracks the author
	Edits preserve the original user_id - add an edited_by field if you need to track who made edits
	For real-time updates, combine with Phoenix PubSub
	Add inserted_at from the event for message timestamps



  

    Event Links Guide

This guide shows how to use Spector's event links feature to create many-to-many relationships between events. Links are useful for tracking ancestry, references, dependencies, and other relationships.
Overview
Event links allow events to reference other events in the same event log. Common use cases include:
	Ancestry tracking: Building tree-structured data like conversation branches
	Edit history: Linking edited versions to their originals
	References: Connecting related events within the same record
	Dependencies: Tracking which events depend on others

Setup
This guide demonstrates typed link tables, which add a type integer column to categorize different relationship types.
Typed Link Schema
Define an Ecto schema to manage the type values:
defmodule MyApp.TypedEventLink do
  use Ecto.Schema
  alias Ecto.Changeset

  @type link_type :: :parent | :sibling | :reference | :supersedes

  @primary_key false
  schema "typed_event_links" do
    field :event_id, :binary_id
    field :linked_id, :binary_id
    # Use explicit integer assignments for forward compatibility.
    # This ensures existing data remains valid if you rename a type.
    field :type, Ecto.Enum, values: [parent: 0, sibling: 1, reference: 2, supersedes: 3]
  end

  def changeset(link \\ %__MODULE__{}, attrs) do
    link
    |> Changeset.cast(attrs, [:event_id, :linked_id, :type])
    |> Changeset.validate_required([:event_id, :linked_id, :type])
  end

  @doc "Create a link struct for insertion"
  def new(event_id, linked_id, type) do
    %__MODULE__{event_id: event_id, linked_id: linked_id, type: type}
  end
end
Document Schema with prepare_event
The prepare_event/3 callback lets you populate link associations before an event is inserted into the event table:
defmodule MyApp.TypedDoc do
  use Spector.Evented, events: MyApp.TypedLinkEvents, actions: [:revise]
  use Ecto.Schema
  alias Ecto.Changeset

  @behaviour Spector.Evented

  @primary_key {:id, :binary_id, autogenerate: false}
  embedded_schema do
    field :content, :string
    field :version, :integer, default: 1
  end

  @impl true
  def prepare_event(event_changeset, previous_events, attrs) do
    case Spector.get_attr(attrs, :supersedes) do
      nil ->
        event_changeset

      superseded_id ->
        # Find the superseded event
        superseded_event = Enum.find(previous_events, &(&1.id == superseded_id))

        if superseded_event do
          # Create a typed link marking this event as superseding another
          link = MyApp.TypedEventLink.new(
            Changeset.get_field(event_changeset, :id),
            superseded_id,
            :supersedes
          )
          Changeset.put_assoc(event_changeset, :links, [link])
        else
          event_changeset
        end
    end
  end

  def changeset(changeset, attrs) when changeset.action == :revise do
    current_version = Changeset.get_field(changeset, :version, 0)

    changeset
    |> Changeset.cast(attrs, [:content])
    |> Changeset.put_change(:version, current_version + 1)
    |> Changeset.validate_required([:content])
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:content])
    |> Changeset.validate_required([:content])
  end

  import Ecto.Query

  def get_superseding_events(event_id) do
    from(link in MyApp.TypedEventLink,
      where: link.linked_id == ^event_id and link.type == :supersedes,
      join: e in MyApp.TypedLinkEvents,
      on: e.id == link.event_id,
      select: e
    )
    |> MyApp.Repo.all()
  end

  def get_superseded_by(event_id) do
    from(link in MyApp.TypedEventLink,
      where: link.event_id == ^event_id and link.type == :supersedes,
      join: e in MyApp.TypedLinkEvents,
      on: e.id == link.linked_id,
      select: e
    )
    |> MyApp.Repo.all()
  end
end
Events Module
defmodule MyApp.TypedLinkEvents do
  use Spector.Events,
    table: "typed_link_events",
    links: [links: {MyApp.TypedEventLink, :linked_id}],
    schemas: [MyApp.TypedDoc],
    repo: MyApp.Repo
end
Usage
Creating a Document
{:ok, doc} = Spector.insert(MyApp.TypedDoc, %{content: "First draft"})
assert doc.content == "First draft"
assert doc.version == 1
Revising with a Supersedes Link
When revising, pass the :supersedes attribute to create a link to the original event:
# Get the original event ID (same as doc.id for insert events)
original_event_id = doc.id

{:ok, revised} = Spector.execute(doc, :revise, %{
  content: "Second draft",
  supersedes: original_event_id
})
assert revised.content == "Second draft"
assert revised.version == 2
Querying Links
Find which events supersede a given event:
superseding = MyApp.TypedDoc.get_superseding_events(original_event_id)
assert length(superseding) == 1
assert hd(superseding).action == :revise
Find what an event supersedes:
import Ecto.Query

# Get the revise event ID from the events table
[revise_event] = MyApp.Repo.all(
  from e in MyApp.TypedLinkEvents,
    where: e.parent_id == ^doc.id and e.action == :revise
)

superseded = MyApp.TypedDoc.get_superseded_by(revise_event.id)
assert length(superseded) == 1
assert hd(superseded).id == original_event_id
Preloading Links
Events with links can be preloaded through the association:
events = MyApp.Repo.all(MyApp.TypedLinkEvents)
  |> MyApp.Repo.preload(:links)

# The revise event should have one link
revise_event = Enum.find(events, &(&1.action == :revise))
assert length(revise_event.links) == 1
assert hd(revise_event.links).type == :supersedes
Link Types Reference
Common relationship types and when to use them:
	Type	Use Case	Example
	:ancestor	Tree-structured history	Conversation branches, version trees
	:parent	Hierarchical relationships	Comment replies, nested items
	:sibling	Peer relationships	Related documents, alternatives
	:reference	Loose connections	Citations, mentions
	:supersedes	Replacement relationships	Edits, corrections, updates
	:depends_on	Dependency tracking	Build steps, prerequisites

Best Practices
	Links must connect events with the same parent_id: Links are designed for relating events within the same record (e.g., linking messages in the same conversation, or revisions of the same document). Do not use links to connect events across different records.

	Choose between typed and separate link tables: Separate link tables (one per relationship type) have a unique index on (event_id, foreign_key). Typed link tables have a unique index on (event_id, foreign_key, type), allowing the same pair of events to have multiple relationships of different types. Typed tables are particularly useful for heterogeneous event tables where you want flexible relationship types without creating a new link table for each one.

	Define clear type semantics: Document what each type means in your domain and enforce it in prepare_event/3.

	Index appropriately: The migration creates indexes on event_id and the foreign key. For typed tables, consider adding a composite index if you frequently query by type.

	Keep links immutable: Links are created when events are inserted and shouldn't be modified afterward. This preserves the integrity of your event log.

	Use Ecto.Enum for types: This provides compile-time checking and clear documentation of valid types.


See Also
	Building an AI Chat Log - Uses links for conversation branching
	Building a Basic Chat - Uses links for edit history



  

    Otzel Integration Guide

This guide covers integrating Otzel, an Operational Transformation (OT) library for Elixir, with Spector event-sourced schemas. Otzel provides Quill.js-compatible rich text operations that pair naturally with event sourcing.
Overview
Why Otzel + Event Sourcing?
Traditional approaches store full document contents on each save. With Otzel and Spector:
	Store diffs, not documents: Events contain delta operations (insertions, deletions, formatting changes) rather than full content
	Automatic composition: When replaying events, deltas compose into the current document state
	Full edit history: Every keystroke or edit is preserved in the event log
	Conflict resolution: OT transforms handle concurrent edits from multiple clients

The Key Insight
Otzel deltas are composable: delta1 |> Otzel.compose(delta2) |> Otzel.compose(delta3) produces the final document state. This maps perfectly to event replay—each event stores a delta, and composing them during replay reconstructs the document.
Basic Usage with Otzel.Ecto.Delta
For simple cases without custom embeds, use Otzel's built-in Ecto type:
defmodule MyApp.Document do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "documents" do
    field :title, :string
    field :content, Otzel.Ecto.Delta, default: Otzel.quill_init()

    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :inserted_at, :updated_at])
    |> apply_delta_change(attrs)
    |> Changeset.validate_required([:title, :content])
  end

  defp apply_delta_change(changeset, attrs) do
    case Spector.fetch_attr(attrs, :content_diff) do
      {:ok, diff} when not is_nil(diff) ->
        changeset
        |> Changeset.get_field(:content)
        |> Otzel.compose(ensure_otzel_structs(diff))
        |> then(&Changeset.put_change(changeset, :content, &1))

      _ ->
        changeset
    end
  end

  # Event payloads come back as JSON maps during replay
  # Convert them back to Otzel structs
  defp ensure_otzel_structs([diff | _] = delta) when not is_struct(diff) do
    {:ok, ops} = Otzel.from_json(delta)
    ops
  end

  defp ensure_otzel_structs(delta), do: delta
end
The Quill Default
Quill.js requires documents to end with a newline character. Use Otzel.quill_init() which returns [Otzel.insert("\n")] as the default value for Quill-backed fields.
Custom Embeds with embed_encoder Option
When you need custom embeds (checkboxes, images, etc.), use the :embed_encoder option with Otzel.from_json/2:
# Signature: Otzel.from_json(json, embed_encoder: {Module, :function})
# Or: Otzel.from_json(json, embed_encoder: &function/1)

# The encoder function receives a raw JSON map and returns:
# - A struct for recognized embeds
# - The original value unchanged for unrecognized data

defmodule MyApp.Embeds do
  def encode(%{"checkbox" => checked}) do
    %MyApp.Checkbox{checked: checked}
  end

  def encode(%{"image" => %{"url" => url, "alt" => alt}}) do
    %MyApp.Image{url: url, alt: alt}
  end

  def encode(other), do: other  # Pass through unchanged
end

# Usage:
{:ok, delta} = Otzel.from_json(json, embed_encoder: {MyApp.Embeds, :encode})

# Or with function capture:
{:ok, delta} = Otzel.from_json(json, embed_encoder: &MyApp.Embeds.encode/1)
Embed Structs
Define struct modules for your custom embeds:
defmodule MyApp.Checkbox do
  @derive Jason.Encoder
  defstruct [:checked]
end

defmodule MyApp.Image do
  @derive Jason.Encoder
  defstruct [:url, :alt]
end
Embeds appear in deltas as Otzel.insert(%MyApp.Checkbox{checked: true}).
Schema with Custom Embeds
When using custom embeds, you need to handle struct conversion during event replay:
defmodule MyApp.Document do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "documents" do
    field :title, :string
    field :content, Otzel.Ecto.Delta, default: Otzel.quill_init()

    # Virtual fields for conflict resolution
    field :last_event_id, :binary_id, virtual: true
    field :replay_events, {:array, :map}, virtual: true

    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :inserted_at, :updated_at])
    |> apply_delta_change(attrs)
    |> Changeset.validate_required([:title, :content])
  end

  defp apply_delta_change(changeset, attrs) do
    case Spector.fetch_attr(attrs, :content_diff) do
      {:ok, diff} when not is_nil(diff) ->
        changeset
        |> Changeset.get_field(:content)
        |> Otzel.compose(ensure_otzel_structs(diff))
        |> then(&Changeset.put_change(changeset, :content, &1))

      _ ->
        changeset
    end
  end

  # Event payloads come back as JSON maps during replay
  # Convert them back to Otzel structs, handling custom embeds
  defp ensure_otzel_structs([diff | _] = delta) when not is_struct(diff) do
    Otzel.from_json(delta, embed_encoder: &embed_encoder/1)
  end

  defp ensure_otzel_structs(delta), do: delta

  defp embed_encoder(%{"checkbox" => checked}) do
    %MyApp.Checkbox{checked: checked}
  end

  defp embed_encoder(%{"image" => %{"url" => url, "alt" => alt}}) do
    %MyApp.Image{url: url, alt: alt}
  end

  defp embed_encoder(other), do: other
end
Delta Composition in Changesets
The core pattern composes incoming diffs with the current document state:
defp apply_delta_change(changeset, attrs) do
  case Spector.fetch_attr(attrs, :content_diff) do
    {:ok, diff} when not is_nil(diff) ->
      changeset
      |> Changeset.get_field(:content)
      |> Otzel.compose(ensure_otzel_structs(diff))
      |> then(&Changeset.put_change(changeset, :content, &1))

    _ ->
      changeset
  end
end
Why Diffs Compose Correctly During Replay
When events replay:
	First event: [] |> compose(diff1) = state after edit 1
	Second event: state1 |> compose(diff2) = state after edit 2
	And so on...

Each event stores only the delta (diff), but composition rebuilds the full document state. This is storage-efficient and provides complete edit history.
Events That Don't Modify the Delta
When an event only modifies non-delta fields (e.g., updating title without changing content), you have two options:
	Omit content_diff: The apply_delta_change function returns the changeset unchanged
	Pass content_diff: []: An empty delta composes to a no-op (Otzel.compose(content, []) == content)

Both approaches work correctly during replay. Omitting the field is cleaner, but explicit [] can be useful for audit trails showing the field was considered.
Warning: Never pass content_diff: nil. The guard when not is_nil(diff) protects against this, but nil values stored in event payloads can cause subtle bugs during replay. Always omit the key entirely or use [].
Conflict Resolution
When multiple clients edit concurrently, the server may receive a diff based on stale state. OT transforms resolve this.
The Pattern
	Client sends diff + previous_id (the last event ID it knew about)
	Server checks if previous_id matches the actual last event
	If stale, transform the incoming diff against intervening server changes
	Store the transformed diff

Full Implementation
defmodule MyApp.Document do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema
  alias Ecto.Changeset

  schema "documents" do
    field :title, :string
    field :content, Otzel.Ecto.Delta, default: Otzel.quill_init()

    # Virtual fields for conflict resolution
    field :last_event_id, :binary_id, virtual: true
    field :replay_events, {:array, :map}, virtual: true

    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Changeset.cast(attrs, [:title, :inserted_at, :updated_at])
    |> apply_delta_change(attrs)
    |> Changeset.validate_required([:title, :content])
  end

  defp apply_delta_change(changeset, attrs) do
    case Spector.fetch_attr(attrs, :content_diff) do
      {:ok, diff} when not is_nil(diff) ->
        diff = ensure_otzel_structs(diff)

        # Check for conflict: did other events happen since the client's base state?
        transformed_diff = maybe_transform_diff(diff, attrs)

        changeset
        |> Changeset.get_field(:content)
        |> Otzel.compose(transformed_diff)
        |> then(&Changeset.put_change(changeset, :content, &1))

      _ ->
        changeset
    end
  end

  defp maybe_transform_diff(diff, attrs) do
    previous_id = Spector.get_attr(attrs, :previous_id)
    replay_events = Spector.get_attr(attrs, :replay_events, [])

    case find_client_base_index(replay_events, previous_id) do
      nil ->
        # No previous_id or not found - no transform needed
        diff

      base_index ->
        # Transform against all events after the client's base
        server_changes =
          replay_events
          |> Enum.drop(base_index + 1)
          |> Enum.map(&extract_content_diff/1)
          |> Enum.reject(&is_nil/1)

        Enum.reduce(server_changes, diff, fn server_diff, client_diff ->
          # Transform: given concurrent edits, how should the client diff be adjusted?
          Otzel.transform(server_diff, client_diff, :right)
        end)
    end
  end

  defp find_client_base_index(events, previous_id) when is_binary(previous_id) do
    Enum.find_index(events, fn event ->
      to_string(event.id) == previous_id or event.id == previous_id
    end)
  end

  defp find_client_base_index(_, _), do: nil

  defp extract_content_diff(%{payload: %{"content_diff" => diff}}) do
    ensure_otzel_structs(diff)
  end

  defp extract_content_diff(%{payload: %{content_diff: diff}}) do
    ensure_otzel_structs(diff)
  end

  defp extract_content_diff(_), do: nil

  defp ensure_otzel_structs([diff | _] = delta) when not is_struct(diff) do
    Otzel.from_json(delta, embed_encoder: &embed_encoder/1)
  end

  defp ensure_otzel_structs(delta), do: delta

  defp embed_encoder(%{"checkbox" => checked}) do
    %MyApp.Checkbox{checked: checked}
  end

  defp embed_encoder(other), do: other
end
Usage with Conflict Detection
def update_document(document, content_diff, previous_id) do
  # Fetch events for replay and conflict detection
  events = Spector.all_events(MyApp.Document, document.id)

  Spector.update(document, %{
    content_diff: content_diff,
    previous_id: previous_id,
    replay_events: events
  })
end
Frontend Integration
On the Elixir side, receive JSON deltas from the frontend and pass to your context:
defmodule MyAppWeb.DocumentLive do
  # Handle delta from Quill.js
  def handle_event("text_change", %{"delta" => delta_json, "previous_id" => prev_id}, socket) do
    case Otzel.from_json(delta_json, embed_encoder: &embed_encoder/1) do
      {:ok, delta} ->
        MyApp.Documents.update_content(
          socket.assigns.document,
          delta,
          prev_id
        )

      {:error, _} ->
        {:noreply, put_flash(socket, :error, "Invalid delta")}
    end
  end

  defp embed_encoder(%{"checkbox" => checked}) do
    %MyApp.Checkbox{checked: checked}
  end

  defp embed_encoder(other), do: other
end
This pattern works with any frontend that sends Quill-compatible delta JSON.
Custom Ecto Type (When Needed)
Create a custom Ecto type only when you need:
	Custom embeds baked into the type
	A different default (non-Quill)

defmodule MyApp.Delta.Richtext do
  use Ecto.Type

  @impl true
  def type, do: :map

  @impl true
  def cast(data) when is_list(data), do: {:ok, data}

  def cast(data) when is_map(data) do
    Otzel.from_json(data, embed_encoder: &embed_encoder/1)
  end

  def cast(_), do: :error

  @impl true
  def load(nil), do: {:ok, Otzel.quill_init()}

  def load(data) when is_map(data) do
    Otzel.from_json(data, embed_encoder: &embed_encoder/1)
  end

  @impl true
  def dump(delta) when is_list(delta), do: {:ok, Otzel.to_json(delta)}
  def dump(_), do: :error

  defp embed_encoder(%{"checkbox" => checked}) do
    %MyApp.Checkbox{checked: checked}
  end

  defp embed_encoder(other), do: other
end
Advanced: Non-Quill OT Fields
Sometimes you need OT for structured data that isn't Quill rich text—annotated sequences, custom formats, or domain-specific structures.
Key Differences from Quill Fields
	Different default: Use [] (empty list), NOT Otzel.quill_init()
	Must be non-null: Use default: [] in the schema
	Different embed encoder: Handle domain-specific embeds

Example: Annotated Sequence
defmodule MyApp.Delta.Sequence do
  use Ecto.Type

  @impl true
  def type, do: :map

  @impl true
  def cast(data) when is_list(data), do: {:ok, data}

  def cast(data) when is_map(data) do
    Otzel.from_json(data, embed_encoder: &sequence_embed_encoder/1)
  end

  def cast(_), do: :error

  @impl true
  def load(nil), do: {:ok, []}  # Empty list, not Quill newline

  def load(data) when is_map(data) do
    Otzel.from_json(data, embed_encoder: &sequence_embed_encoder/1)
  end

  @impl true
  def dump(delta) when is_list(delta), do: {:ok, Otzel.to_json(delta)}
  def dump(_), do: :error

  # Domain-specific embeds
  defp sequence_embed_encoder(%{"marker" => %{"type" => type, "position" => pos}}) do
    %MyApp.Marker{type: type, position: pos}
  end

  defp sequence_embed_encoder(other), do: other
end
defmodule MyApp.Annotation do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "annotations" do
    field :name, :string
    field :sequence, MyApp.Delta.Sequence, default: []  # Empty list default
    timestamps()
  end
end
Multiple OT Types in One Application
You may need different OT field types in the same application.
Pattern 1: Quill + Non-Quill Fields
When you have both a rich text editor and a structured OT field:
defmodule MyApp.AnnotatedDocument do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "annotated_documents" do
    # Quill rich text - requires trailing newline
    field :content, Otzel.Ecto.Delta, default: Otzel.quill_init()

    # Structured annotations - empty list default
    field :annotations, MyApp.Delta.Sequence, default: []

    timestamps()
  end
end
Each field uses its own Ecto type with the appropriate:
	Default value (Otzel.quill_init() vs [])
	Embed encoder (Quill embeds vs domain embeds)

Pattern 2: Multiple Quill Fields with Different Plugins
When you have multiple Quill editors with different embed configurations, create custom types:
defmodule MyApp.Delta.BasicRichtext do
  use Ecto.Type
  # Standard Quill with no custom embeds
  # ... standard implementation using Otzel.from_json/1
end

defmodule MyApp.Delta.ChecklistRichtext do
  use Ecto.Type
  # Quill with checkbox support
  defp embed_encoder(%{"checkbox" => checked}) do
    %MyApp.Checkbox{checked: checked}
  end
  defp embed_encoder(other), do: other
end
Best Practices
	Use Otzel.Ecto.Delta for simple cases: Only create custom Ecto types when you need custom embeds baked into the type

	Use the correct embed_encoder signature:
# Correct - as keyword option
Otzel.from_json(data, embed_encoder: &embed_encoder/1)
Otzel.from_json(data, embed_encoder: {Module, :function})

# Wrong - as positional argument
Otzel.from_json(data, &embed_encoder/1)

	Embed encoders return values directly (not wrapped in {:ok, ...}):
# Correct
defp embed_encoder(%{"checkbox" => checked}) do
  %MyApp.Checkbox{checked: checked}
end
defp embed_encoder(other), do: other

# Wrong
defp embed_encoder(%{"checkbox" => checked}) do
  {:ok, %MyApp.Checkbox{checked: checked}}
end

	Always store diffs, not full documents: Pass content_diff to your changeset, compose with current state

	Use the correct default:
	Quill fields: Otzel.quill_init() (returns [Otzel.insert("\n")])
	Non-Quill OT fields: []


	Handle JSON round-trip in changesets: Event payloads come back as JSON maps during replay. Use ensure_otzel_structs pattern with Otzel.from_json/2

	Handle both atom and string keys: Use Spector.get_attr/2 or Spector.fetch_attr/2

	Test conflict resolution: Write tests that simulate concurrent edits with stale previous_id values


See Also
	Otzel Documentation
	Quill.js Delta Format
	Building a Basic Chat - For simpler text fields without OT

CQRS-style event sourcing for Ecto schemas.
Spector records all changes to your Ecto schemas as events in a separate event
log table. This enables full audit trails, temporal queries, and the ability to
replay history. For tamper-evident logs, enable optional hash chain integrity.
Quick Start
1. Define an Events Table
defmodule MyApp.Events do
  use Spector.Events,
    table: "events",
    schemas: [MyApp.User, MyApp.Post],
    repo: MyApp.Repo
end
2. Mark Schemas as Evented
defmodule MyApp.User do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "users" do
    field :name, :string
    field :email, :string
  end

  def changeset(changeset, attrs) do
    changeset
    |> Ecto.Changeset.cast(attrs, [:name, :email])
    |> Ecto.Changeset.validate_required([:name, :email])
  end
end
3. Create Migrations
defmodule MyApp.Repo.Migrations.CreateEvents do
  use Ecto.Migration

  def up, do: Spector.Migration.up(table: "events")
  def down, do: Spector.Migration.down(table: "events")
end
4. Use Spector Instead of Repo
# Insert
{:ok, user} = Spector.insert(MyApp.User, %{name: "Alice", email: "alice@example.com"})

# Update
{:ok, user} = Spector.update(user, %{name: "Alice Smith"})

# Delete
{:ok, user} = Spector.delete(user)
How It Works
When you update or execute an action on a record, Spector "rolls forward" by
replaying all stored events through your schema's changeset/2 function. This means:
	Your changeset function handles both new operations AND historical replay
	Schema migrations happen automatically during replay (using version guards)
	The current state is always reconstructed from the event log
	Stale in-memory records are never a problem

This design lets you evolve your schema over time while maintaining full
compatibility with historical events.
Guides
For complete examples of building applications with Spector:
	Building an AI Chat Log - Conversation branching with tree-structured message history
	Associations Guide - Handling Ecto associations with event-sourced schemas
	Building a Basic Chat - Simple chat with edit history tracking
	Event Links Guide - Creating many-to-many relationships between events
	Otzel Integration - Rich text editing with Operational Transformation

Features
Custom Actions
Define domain-specific actions beyond insert/update/delete:
defmodule MyApp.Item do
  use Spector.Evented, events: MyApp.Events, actions: [:archive]

  def changeset(changeset, attrs) when changeset.action == :archive do
    Ecto.Changeset.change(changeset, archived_at: Spector.get_attr(attrs, :archived_at))
  end

  def changeset(changeset, attrs) do
    Ecto.Changeset.cast(changeset, attrs, [:name, :value])
  end
end

# Execute custom action
{:ok, item} = Spector.execute(item, :archive, %{archived_at: DateTime.utc_now()})
Schema Versioning
Handle schema migrations with version guards:
defmodule MyApp.User do
  use Spector.Evented, events: MyApp.Events, version: 1

  # Migrate v0 events (with :title) to v1 (with :name)
  def changeset(changeset, attrs) when version_is(attrs, 0) do
    attrs = Map.put(attrs, "name", Spector.get_attr(attrs, :title))
    do_changeset(changeset, attrs)
  end

  def changeset(changeset, attrs), do: do_changeset(changeset, attrs)
end
Hash Chain Integrity
Enable tamper-evident event logs with cryptographic hashing:
defmodule MyApp.Events do
  use Spector.Events,
    table: "events",
    schemas: [MyApp.User],
    repo: MyApp.Repo,
    hashed: true
end
Explicit Schema Indexing
Ensure stability when adding/removing schemas:
schemas: [MyApp.User, MyApp.Post, {MyApp.Comment, 10}]
Action Aliases
Maintain backwards compatibility when renaming actions:
use Spector.Events,
  aliases: [soft_delete: :archive]
Reserved Attributes
Spector injects reserved attributes into the attrs map passed to your
changeset/2 function. These provide metadata about the event being applied:
	:__version__ - The schema version when the event was created. Use with
version_is/2 guards to handle schema migrations during replay.

	:__event_id__ - The unique ID of the event being applied. Use
Spector.changeset_put_event_id/3 to assign this to a field:
def changeset(message, attrs) do
  message
  |> Ecto.Changeset.cast(attrs, [:content])
  |> Spector.changeset_put_event_id(attrs)  # puts :__event_id__ into :id field
end


These attributes are also stored in the event payload for reference.
Database Support
Spector works with any database supported by Ecto for basic functionality.
Note: Hashed event tables (hashed: true) currently require PostgreSQL.
The hash chain integrity feature uses LOCK TABLE ... IN EXCLUSIVE MODE which
is PostgreSQL-specific.

          @spec all_events(evented_schema(), Ecto.UUID.t()) :: [struct()]


      


Returns all events for a record from the beginning.
Events are returned in insertion order.
Example
events = Spector.all_events(MyApp.User, user_id)

          @spec bringup(evented_schema(), bringup_opts()) :: {:ok, [evented_struct()]}


      


Import existing database records into the event log.
Reads all rows from the schema's table and migrates each to Spector management:
deletes the original record and creates a new one with a UUIDv7 ID and
corresponding event. The entire operation runs in a single transaction.
Records that are already tracked by Spector (have existing events) are skipped.
Options
	:action - The action to use for the event (default: :insert). Use a custom
action like :import to trigger different changeset behavior during migration.

	:attr_fn - A function that takes a record and returns the attributes map to
insert (default: extracts all schema fields except the primary key). Use this to
transform or augment data during migration.

	:transfer - A function that receives the old record and the new record before
the old record is deleted. Use this to update associations or perform other
transfer operations (default: no-op).


Returns {:ok, [struct]} on success or {:error, reason} on failure.
Timestamps
By default, bringup creates new records, so inserted_at and updated_at timestamps
will be set to the current time. To preserve original timestamps from the source
records, include them in the attr_fn and ensure your changeset accepts them.
If you don't want your regular changeset to accept timestamp fields, use a custom
action like :import to handle them separately:
# Register :import as a custom action
use Spector.Evented, events: MyApp.Events, actions: [:import]

# Handle :import with timestamp support
def changeset(changeset, attrs) when changeset.action == :import do
  changeset
  |> cast(attrs, [:name, :inserted_at, :updated_at])
  |> validate_required([:name])
end

# Regular changeset doesn't accept timestamps
def changeset(changeset, attrs) do
  changeset
  |> cast(attrs, [:name])
  |> validate_required([:name])
end

# Pass timestamps in attr_fn
attr_fn = fn record ->
  %{name: record.name, inserted_at: record.inserted_at, updated_at: record.updated_at}
end

{:ok, users} = Spector.bringup(MyApp.User, action: :import, attr_fn: attr_fn)
Examples
Basic usage migrates all untracked records (timestamps reset to now):
{:ok, users} = Spector.bringup(MyApp.User)
Use a custom attr_fn to transform data during migration:
{:ok, users} = Spector.bringup(MyApp.User, attr_fn: fn record ->
  %{
    name: String.upcase(record.name),
    value: record.value || 0  # provide defaults for nil values
  }
end)
Use a transfer function to update associations before the old record is deleted:
{:ok, users} = Spector.bringup(MyApp.User, transfer: fn old, new ->
  Repo.update_all(
    from(p in Post, where: p.user_id == ^old.id),
    set: [user_id: new.id]
  )
end)

          @spec changeset_put_event_id(Ecto.Changeset.t(), attrs(), atom()) ::
  Ecto.Changeset.t()


      


Assigns the current event ID to a changeset field.
When Spector calls your changeset/2 function, it includes an :__event_id__
key in the attrs map. This function extracts that ID and assigns it to
the specified field in your changeset.
This is useful for embedded schemas and {:array, :map} rollup records
where you want each record to have a unique ID that matches its creation event.
Parameters
	changeset - The changeset to modify
	attrs - The attrs map passed to changeset/2 (contains :__event_id__)
	field - The field to assign the event ID to (default: :id)

Example
def changeset(message, attrs) do
  message
  |> Ecto.Changeset.cast(attrs, [:content, :role])
  |> Spector.changeset_put_event_id(attrs)
  |> Ecto.Changeset.validate_required([:id, :content, :role])
end

          @spec delete(evented_struct()) ::
  {:ok, evented_struct()} | {:error, Ecto.Changeset.t()}


      


Delete a record, creating a delete event in the event log.
Returns {:ok, struct} on success or {:error, changeset} on failure.
The delete event is recorded in the event log before the record is removed
from the database, providing a complete audit trail.
Example
{:ok, user} = Spector.delete(user)

          @spec execute(evented_struct(), action(), attrs()) ::
  {:ok, evented_struct()} | {:error, Ecto.Changeset.t()}


      


Execute an action on a record, creating an event in the event log.
This is the general-purpose function for applying any action to a record,
including custom actions defined in the schema's :actions option.
Returns {:ok, struct} on success or {:error, changeset} on failure.
The function rolls forward from the event log to reconstruct current state,
applies the action through the schema's changeset/2 function, and records
the new event.
Example
# Using a custom :archive action
{:ok, item} = Spector.execute(item, :archive, %{archived_at: DateTime.utc_now()})

# The :update action (same as Spector.update/2)
{:ok, user} = Spector.execute(user, :update, %{name: "New Name"})

          @spec fetch_attr(attrs(), atom()) :: {:ok, any()} | :error


      


Fetches a field from attrs, checking both atom and string keys.
Returns {:ok, value} if the key exists, or :error if not found.
This is useful in changesets where attrs may come with string keys (from JSON)
or atom keys (from internal calls).
Example
def changeset(record, attrs) do
  case Spector.fetch_attr(attrs, :parent_id) do
    {:ok, parent_id} -> # handle parent_id
    :error -> # no parent_id provided
  end
end

          @spec fetch_attr!(attrs(), atom()) :: any()


      


Fetches a field from attrs, checking both atom and string keys.
Returns the value if the key exists, or raises KeyError if not found.
Example
def changeset(record, attrs) do
  parent_id = Spector.fetch_attr!(attrs, :parent_id)
  # use parent_id
end

          @spec get(evented_schema(), String.t()) :: evented_struct() | nil


      


Retrieve the current state of a record by replaying its events.
Returns the struct if found, or nil if no events exist for the given ID.
This is useful for embedded schemas (without database tables) or when you
want to reconstruct state purely from the event log.
Example
user = Spector.get(MyApp.User, "019ac640-dfc0-7407-8238-39a9c45e8813")

          @spec get_attr(attrs(), atom(), any()) :: any()


      


Gets a field from attrs, checking both atom and string keys.
Returns the value if the key exists, or default if not found.
Example
def changeset(record, attrs) do
  parent_id = Spector.get_attr(attrs, :parent_id, nil)
  # use parent_id, which may be nil
end

          @spec insert(evented_schema(), attrs(), action()) ::
  {:ok, evented_struct()} | {:error, Ecto.Changeset.t()}


      


Insert a new record, creating an event in the event log.
Returns {:ok, struct} on success or {:error, changeset} on failure.
Example
{:ok, user} = Spector.insert(MyApp.User, %{name: "Alice", email: "alice@example.com"})
The inserted struct will have a new UUIDv7 id assigned.

          @spec materialize(module(), Ecto.UUID.t()) ::
  {:ok, evented_struct()} | {:error, :invalid | :deleted | Ecto.Changeset.t()}


      


Materialize a record from the event log into its database table.
Takes an events module and parent_id, replays all events to reconstruct the
record, and inserts it into the database.
Returns {:ok, struct} on success, or an error tuple:
	{:error, :invalid} - No events exist for the given parent_id
	{:error, :deleted} - Event history ends with a delete action
	{:error, changeset} - Insert failed (e.g., record already exists)

Example
{:ok, user} = Spector.materialize(MyApp.Events, parent_id)

Returns all events up to and including the given event.
Events are returned in insertion order. The specified event is included
as the last element in the result.
Note
This function does not take into account savepoints.
Example
events = Spector.previous_events(event)

  



  
    
      
    
    
      recent_events(schema, parent_id)



        
          
        

    

  


  

      

          @spec recent_events(evented_schema(), Ecto.UUID.t()) :: [struct()]


      


Returns events starting from the most recent savepoint.
If no savepoint exists, returns all events from the beginning.
The savepoint event itself is included as the first element.
Events are returned in insertion order.
Use of this function over all_events/2 is preferable; if the
schema does not support savepoints, the two functions behave
identically.
Example
events = Spector.recent_events(MyApp.User, user_id)

Create a savepoint event capturing the current state of a record.
Returns {:ok, struct} on success or raises on failure.
Savepoints store the complete state of a record at a point in time, allowing
event replay to start from the savepoint instead of replaying all events
from the beginning. This is useful for records with long event histories.
The schema must implement the savepoint/2 callback to define how the
current state is converted to an attrs map:
@behaviour Spector.Evented

@impl true
def savepoint(record, _version) do
  %{name: record.name, email: record.email}
end
Forms
There are two ways to create a savepoint:
From a record (savepoint/1)
Pass the record directly. This verifies that the record matches the current
state in the event log (replayed events must produce the same field values).
This guards against creating savepoints from stale records:
{:ok, user} = Spector.savepoint(user)
If the record is stale (e.g., another process updated it), this raises an error.
From schema and ID (savepoint/2)
Pass the schema module and record ID. This replays events to determine
current state without verification:
{:ok, user} = Spector.savepoint(MyApp.User, user_id)
Use this form when you don't have the record in memory or don't need
stale record detection.

          @spec update(evented_struct(), attrs()) ::
  {:ok, evented_struct()} | {:error, Ecto.Changeset.t()}


      


Update an existing record, creating an event in the event log.
This is a convenience function that calls execute(record, :update, attrs).
Returns {:ok, struct} on success or {:error, changeset} on failure.
Example
{:ok, user} = Spector.update(user, %{name: "Alice Smith"})

Mark a schema as evented, linking it to an event log table.
Basic Usage
defmodule MyApp.User do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "users" do
    field :name, :string
    timestamps()
  end

  def changeset(changeset, attrs) do
    changeset
    |> Ecto.Changeset.cast(attrs, [:name, :inserted_at, :updated_at])
    |> Ecto.Changeset.validate_required([:name])
  end
end
Timestamps in Changesets
Always include inserted_at and updated_at in your changeset's cast fields
when using timestamps(). Spector stores these values in event payloads to
ensure proper timestamp replay. Without casting them, replayed events won't
restore original timestamps.
Options
	:events (required) - The events module (defined with Spector.Events)
	:repo - Override the repo for this schema's table (default: uses events repo)
	:version - Schema version for migrations (default: 0). See "Schema Versioning" below
	:actions - List of custom action atoms (default: []). See "Custom Actions" below

Custom Actions
Beyond the built-in :insert, :update, and :delete actions, you can define
custom actions for domain-specific operations:
defmodule MyApp.Item do
  use Spector.Evented,
    events: MyApp.Events,
    actions: [:archive, :restore]

  schema "items" do
    field :name, :string
    field :value, :integer
    field :archived_at, :utc_datetime_usec
    timestamps()
  end

  # Handle the archive action
  def changeset(changeset, attrs) when changeset.action == :archive do
    changeset
    |> Ecto.Changeset.change(archived_at: Spector.get_attr(attrs, :archived_at))
  end

  # Handle other actions
  def changeset(changeset, attrs) do
    changeset
    |> Ecto.Changeset.cast(attrs, [:name, :value, :inserted_at, :updated_at])
  end
end
Execute custom actions with Spector.execute/3:
{:ok, item} = Spector.execute(item, :archive, %{archived_at: DateTime.utc_now()})
Schema Versioning
When you change your schema (add/remove/rename fields), increment the version
and handle migrations in your changeset:
defmodule MyApp.User do
  # Version 0: had :title field
  # Version 1: renamed :title to :name
  use Spector.Evented, events: MyApp.Events, version: 1

  schema "users" do
    field :name, :string
    timestamps()
  end

  # Migrate v0 events (with :title) to v1 (with :name)
  def changeset(changeset, attrs) when version_is(attrs, 0) do
    attrs = Map.put(attrs, "name", Spector.get_attr(attrs, :title))
    do_changeset(changeset, attrs)
  end

  def changeset(changeset, attrs), do: do_changeset(changeset, attrs)

  defp do_changeset(changeset, attrs) do
    changeset
    |> Ecto.Changeset.cast(attrs, [:name, :inserted_at, :updated_at])
    |> Ecto.Changeset.validate_required([:name])
  end
end
The version_is/2 and version_in/2 guards help you handle different versions.
During updates, Spector replays all stored events through your changeset/2
function. Old events retain their original version, so your version guards
automatically migrate historical data during replay.
Generated Functions
Using this module generates:
	__spector__/1 - Internal metadata accessor
	Imports version_in/2 and version_is/2 guards
	Sets @primary_key to {:id, UUIDv7, autogenerate: false} (Spector manages IDs)

Customizing the Primary Key
The default @primary_key can be overridden by defining it after use Spector.Evented:
defmodule MyApp.Record do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  @primary_key {:uuid, UUIDv7, autogenerate: false}
  schema "records" do
    field :name, :string
  end
end
The primary key must be a binary type (not integer). This is validated at compile time.
Optional Callbacks
prepare_event/3
Called before inserting an event, allowing last-minute modifications to the
event changeset. Useful for populating link associations.
@behaviour Spector.Evented

@impl true
def prepare_event(event_changeset, existing_events, attrs) do
  # Modify event_changeset as needed
  event_changeset
end
The callback receives:
	event_changeset - The changeset for the event about to be inserted
	existing_events - All existing events for this record (with links preloaded)
	attrs - The attributes passed to the action

Note: prepare_event/3 always runs inside a transaction.

        prepare_event(event_changeset, existing_events, attrs)

      


    


    
      
        prepare_materialization(changeset)

      


        Prepare the final changeset before materialization.



    


    
      
        savepoint(record, version)

      


        Convert the current record state into an attrs map for a savepoint event.



    





  
    Functions
  


    
      
        event_log(name)

      


        Creates a has_many association to the event log for this record.



    


    
      
        version_in(attrs, range)

      


        Guard to check if attrs version is within a range or list of integers.



    


    
      
        version_is(attrs, version)

      


        Guard to check if attrs version equals a specific integer.

      prepare_event(event_changeset, existing_events, attrs)


        (optional)


        
          
        

    

  


  

      

          @callback prepare_event(
  event_changeset :: Ecto.Changeset.t(),
  existing_events :: [struct()],
  attrs :: map()
) :: Ecto.Changeset.t()


      



  



  
    
      
    
    
      prepare_materialization(changeset)


        (optional)


        
          
        

    

  


  

      

          @callback prepare_materialization(changeset :: Ecto.Changeset.t()) :: Ecto.Changeset.t()


      


Prepare the final changeset before materialization.
Called on the accumulated changeset after all events have been replayed,
just before Repo.insert() or Repo.update(). Use this to convert
accumulated data into associations.
Runs inside the same transaction as the insert/update.
This callback is primarily intended for database-backed schemas. For embedded
schemas, it is not called automatically, but you may define it for manual use
or to enforce invariants.
Example
@impl Spector.Evented
def prepare_materialization(changeset) do
  assignee_ids = Changeset.get_field(changeset, :assignee_ids) || []
  users = Enum.map(assignee_ids, &%MyApp.User{id: &1})
  Changeset.put_assoc(changeset, :assignees, users)
end

  



  
    
      
    
    
      savepoint(record, version)


        (optional)


        
          
        

    

  


  

      

          @callback savepoint(record :: struct(), version :: non_neg_integer()) :: map()


      


Convert the current record state into an attrs map for a savepoint event.
Called by Spector.savepoint/2 to capture the full state of a record.
When replaying events, savepoints allow starting from an intermediate state
instead of replaying from the beginning.
The version parameter is the schema version at the time the savepoint was
registered (from the savepoint event's __version__ field). This allows
handling schema migrations when replaying from older savepoints.
@behaviour Spector.Evented

@impl true
def savepoint(record, _version) do
  %{
    name: record.name,
    email: record.email,
    status: record.status
  }
end
The returned attrs map should contain all fields needed to reconstruct
the record's state at this point.
Testing savepoints is highly encouraged. Use Spector.Integrity.verify_savepoints/2
to ensure savepoints correctly capture the replayed state from all possible replay paths.

  


        

      

      
        Functions


        


  
    
      
    
    
      event_log(name)


        (macro)


        
          
        

    

  


  

Creates a has_many association to the event log for this record.
Use this inside your schema definition to add an association that retrieves
all events for a given record.
Note: This macro only works with database-backed schemas, not embedded schemas.
Embedded schemas cannot use Ecto associations for preloading.
Example
defmodule MyApp.User do
  use Spector.Evented, events: MyApp.Events
  use Ecto.Schema

  schema "users" do
    field :name, :string
    event_log :log
  end
end
Then you can preload and access events:
user = Repo.get(User, id) |> Repo.preload(:log)
user.log  # Returns all events for this user

  



  
    
      
    
    
      version_in(attrs, range)


        (macro)


        
          
        

    

  


  

Guard to check if attrs version is within a range or list of integers.
Example: def changeset(struct, attrs) when version_in(attrs, 0..2)

  



  
    
      
    
    
      version_is(attrs, version)


        (macro)


        
          
        

    

  


  

Guard to check if attrs version equals a specific integer.
Example: def changeset(struct, attrs) when version_is(attrs, 0)

Define an event log table for storing events from evented schemas.
Basic Usage
defmodule MyApp.Events do
  use Spector.Events,
    table: "events",
    schemas: [MyApp.User, MyApp.Post],
    repo: MyApp.Repo
end
Options
	:table (required) - The database table name for storing events
	:schemas (required) - List of schemas that will log events to this table
	:repo (required) - The Ecto repo module to use for database operations
	:hashed - Enable hash chain integrity (default: false). See "Hash Chain Integrity" below
	:aliases - Action aliases for refactoring. See "Action Aliases" below
	:shard - Sharding function name (atom). See "Table Sharding" below
	:links - List of link associations for many-to-many relationships. See "Event Links" below

Table Sharding
For high-volume event logs, you can shard events across multiple tables using
the :shard option. Specify a function name (atom) that takes a parent_id
and returns the table name:
defmodule MyApp.Events do
  use Spector.Events,
    table: "events",  # base table name (used for schema definition)
    schemas: [MyApp.User],
    repo: MyApp.Repo,
    shard: :shard_table

  def shard_table(parent_id) do
    # Shard based on first byte of UUID
    <<first_byte, _rest::binary>> = parent_id
    "events_#{rem(first_byte, 4)}"
  end
end
You'll need to create migrations for each shard table. The sharding function
must be deterministic - the same parent_id must always map to the same table.
When combining sharding with hash chain integrity (hashed: true), each shard
table maintains its own independent hash chain.
Event Links
Enable event linking with the :links option to create many-to-many relationships
between events:
use Spector.Events,
  table: "events",
  schemas: [MyApp.Chat],
  repo: MyApp.Repo,
  links: [ancestors: {"event_ancestors", :ancestor_id}]
Each link creates a join table and a many_to_many association on the events module.
You can define multiple links:
links: [
  ancestors: {"event_ancestors", :ancestor_id},
  categories: {"event_categories", :category_id}
]
For links that use an Ecto schema module, pass the module instead of a table name:
links: [links: {MyApp.EventLink, :linked_id}]
The two approaches differ in what the association returns:
	Table name string: Creates a many_to_many association. Preloading returns the
linked events directly. You cannot use put_assoc to set fields on the join table.

	Schema module: Creates a has_many association. Preloading returns the link
records themselves (e.g., %EventLink{event_id: ..., linked_id: ..., type: ...}).
You can use put_assoc in prepare_event/3 to create links with custom fields.


Use a schema module when you need to store metadata on links (types, timestamps, etc.)
or need to create links via put_assoc. Use a table name string for simple joins where
you only care about which events are connected.
See the Event Links Guide for details.
Sharding with Links
When combining sharding with links, a separate link table is created for each shard.
Use link_table_for/2 to get the correct link table name for a given parent_id:
def prepare_event(event_changeset, _previous_events, attrs) do
  parent_id = Ecto.Changeset.get_field(event_changeset, :parent_id)
  link_table = MyEvents.link_table_for(parent_id, "ancestors")

  link = %MyLink{ancestor_id: attrs[:ancestor_id]}
  link = Ecto.put_meta(link, source: link_table)

  Ecto.Changeset.put_assoc(event_changeset, :ancestors, [link])
end
Schema Indexing
Schemas are stored as integers in the database. By default, schemas are
auto-indexed starting from 0. To ensure stability when adding/removing schemas,
you can specify explicit indexes:
schemas: [MyApp.User, MyApp.Post, {MyApp.Comment, 10}]
In this example, User gets index 0, Post gets index 1, and Comment gets
index 10. This allows you to remove Post later without breaking existing data.
Hash Chain Integrity
Enable hashed: true to create a cryptographic hash chain linking all events:
use Spector.Events,
  table: "events",
  schemas: [MyApp.User],
  repo: MyApp.Repo,
  hashed: true
Each event's hash includes the previous event's hash, creating a tamper-evident
chain. Any modification to historical events will break the chain.
Note: Hash chain integrity requires PostgreSQL due to the use of
LOCK TABLE ... IN EXCLUSIVE MODE for serialization.
Action Aliases
When refactoring action names, use aliases to maintain backwards compatibility
with existing events in the database:
use Spector.Events,
  table: "events",
  schemas: [MyApp.Item],
  repo: MyApp.Repo,
  aliases: [soft_delete: :archive]  # soft_delete uses archive's hash
This allows renaming :archive to :soft_delete in your code while still
reading old events that used :archive.
Compilation Dependencies
Using this module creates a compilation dependency on all schema modules
listed in the :schemas option. This means changes to those schema modules
will trigger recompilation of the events module.
Generated Functions
Using this module generates the following functions:
	changeset/1, changeset/2 - Build an event changeset
	table_for/1 - Get the table name for a given parent_id
	shard/2 - Apply sharding to a changeset
	__spector__/1 - Internal metadata accessor


      


      
        Summary


  
    Callbacks
  


    
      
        changeset(attrs)

      


        Build an event changeset from attributes.



    


    
      
        changeset(struct, attrs)

      


        Build an event changeset from an existing struct and attributes.



    


    
      
        link_table_for(parent_id, link_table)

      


        Get the link table name for a given parent_id and base link table name.



    


    
      
        shard(changeset, parent_id)

      


        Apply sharding to a changeset based on the parent_id.



    


    
      
        table_for(parent_id)

      


        Get the table name for a given parent_id.

      changeset(attrs)



        
          
        

    

  


  

      

          @callback changeset(attrs :: map()) :: Ecto.Changeset.t()


      


Build an event changeset from attributes.

  



  
    
      
    
    
      changeset(struct, attrs)



        
          
        

    

  


  

      

          @callback changeset(struct :: struct(), attrs :: map()) :: Ecto.Changeset.t()


      


Build an event changeset from an existing struct and attributes.

  



  
    
      
    
    
      link_table_for(parent_id, link_table)



        
          
        

    

  


  

      

          @callback link_table_for(parent_id :: binary(), link_table :: String.t()) :: String.t()


      


Get the link table name for a given parent_id and base link table name.
With sharding, link tables are prefixed with the shard table name.
Without sharding, returns the link table name unchanged.

  



  
    
      
    
    
      shard(changeset, parent_id)



        
          
        

    

  


  

      

          @callback shard(changeset :: Ecto.Changeset.t(), parent_id :: binary()) ::
  Ecto.Changeset.t()


      


Apply sharding to a changeset based on the parent_id.
Updates the changeset's data source to the appropriate shard table.

  



  
    
      
    
    
      table_for(parent_id)



        
          
        

    

  


  

      

          @callback table_for(parent_id :: binary()) :: String.t()


      


Get the table name for a given parent_id.
For non-sharded tables, returns the configured table name.
For sharded tables, calls the shard function to determine the table.

  


        

      


  

    
Spector.Integrity 
    



      
Integrity verification for Spector event logs.
This module provides functions to verify the integrity of event logs,
including savepoint validation and hash chain verification.

      


      
        Summary


  
    Functions
  


    
      
        verify_hash_chain(events_module)

      


        Verify the hash chain integrity of an entire events table.



    


    
      
        verify_savepoints(schema, parent_id)

      


        Verify all savepoints for a record match the expected state at that point.

      verify_hash_chain(events_module)



        
          
        

    

  


  

      

          @spec verify_hash_chain(module()) ::
  :ok | {:error, Spector.Integrity.HashMismatch.t()}


      


Verify the hash chain integrity of an entire events table.
Iterates through all events in the table ordered by inserted_at and verifies
that each event's hash correctly chains from the previous event.
Returns :ok if the hash chain is valid, or {:error, exception} where
exception is a Spector.Integrity.HashMismatch.
Raises if the events module doesn't have hash chain integrity enabled.
Examples
# Verify the hash chain for an events table
:ok = Spector.Integrity.verify_hash_chain(MyApp.Events)

# Handle verification failures
case Spector.Integrity.verify_hash_chain(MyApp.Events) do
  :ok -> :verified
  {:error, %Spector.Integrity.HashMismatch{} = failure} ->
    Logger.error(Exception.message(failure))
end

  



  
    
      
    
    
      verify_savepoints(schema, parent_id)



        
          
        

    

  


  

      

          @spec verify_savepoints(module(), String.t()) ::
  :ok | {:error, Spector.Integrity.SavepointFailure.t()}


      


Verify all savepoints for a record match the expected state at that point.
Replays events from the beginning up to each savepoint and verifies that
all replay paths (skipping different combinations of savepoints) converge
to the same state. This catches bugs in savepoint/2 implementations that
omit fields.
Returns :ok if all savepoints are valid, or {:error, exception} where
exception is a Spector.Integrity.SavepointFailure.
Raises if the schema doesn't implement the savepoint/2 callback.
Examples
# Verify all savepoints for a record
:ok = Spector.Integrity.verify_savepoints(MyApp.User, user_id)

# Handle verification failures
case Spector.Integrity.verify_savepoints(MyApp.User, user_id) do
  :ok -> :verified
  {:error, %Spector.Integrity.SavepointFailure{} = failure} ->
    Logger.error(Exception.message(failure))
end

Migration helpers for creating Spector event tables.
Basic Usage
defmodule MyApp.Repo.Migrations.CreateEvents do
  use Ecto.Migration

  def up, do: Spector.Migration.up(table: "events")
  def down, do: Spector.Migration.down(table: "events")
end
Options
	:table (required) - The database table name for the events
	:shards - List of table names for sharded setups (creates multiple tables)
	:hashed - Add a hash column for hash chain integrity (default: false)
	:links - List of link tables to create: [{"table_name", :foreign_key}, ...]

With Table Sharding
For sharded event tables, use :shards to create multiple tables:
def up do
  Spector.Migration.up(shards: ["events_0", "events_1", "events_2", "events_3"])
end

def down do
  Spector.Migration.down(shards: ["events_0", "events_1", "events_2", "events_3"])
end
When combining shards with links, a link table is created for each shard by
concatenating the shard name to the link table name. For example:
Spector.Migration.up(
  shards: ["events_0", "events_1"],
  links: [{"ancestors", :ancestor_id}]
)
This creates four tables: events_0, events_1, events_0_ancestors, and
events_1_ancestors. If you need a different naming convention, omit the
:links option and use link_up/2 separately for each shard.
With Hash Chain Integrity
If using hashed events, include the :hashed option:
def up, do: Spector.Migration.up(table: "events", hashed: true)
def down, do: Spector.Migration.down(table: "events")
With Event Links
To create join tables for event linking (e.g., ancestry tracking):
def up do
  Spector.Migration.up(
    table: "events",
    links: [{"event_ancestors", :ancestor_id}]
  )
end

def down do
  Spector.Migration.down(
    table: "events",
    links: [{"event_ancestors", :ancestor_id}]
  )
end
Each link tuple creates a join table with event_id and the specified foreign key,
along with indexes for efficient queries. A database trigger enforces that both
ends of a link must have the same parent_id (i.e., links can only connect events
within the same record).
Typed Link Tables
To distinguish different relationship types on the same link table, use the
:typed option to add a type integer column:
def up do
  Spector.Migration.up(
    table: "events",
    links: [{"event_links", :linked_id, typed: true}]
  )
end
The type column allows a single link table to represent multiple relationship types
(e.g., "parent", "sibling", "reference") by assigning each type an integer value.
It is recommended to use an Ecto schema for typed link tables. See the
Event Links Guide for examples.
Generated Schema
The migration creates a table with:
	id - UUIDv7 primary key
	parent_id - Foreign key reference to the first event (self-referential)
	payload - Map/JSONB column storing the event data
	schema - Integer identifying which schema this event belongs to
	action - Integer identifying the action (insert, update, delete, or custom)
	hash - Binary column for SHA-256 hash (only if hashed: true)
	inserted_at, updated_at - Timestamps with microsecond precision

Indexes are created on schema and parent_id for efficient queries.

      


      
        Summary


  
    Functions
  


    
      
        down(opts)

      


        Drop the event table(s) and any link tables.



    


    
      
        link_down(arg)

      


        Drop a link table.



    


    
      
        link_up(events_table, arg)

      


        Create a link table for many-to-many event relationships.



    


    
      
        up(opts)

      


        Create the event table(s) and any link tables.

      down(opts)



        
          
        

    

  


  

Drop the event table(s) and any link tables.
Pass the same options used in up/1 to ensure link tables are also dropped.

  



  
    
      
    
    
      link_down(arg)



        
          
        

    

  


  

Drop a link table.
Example
Spector.Migration.link_down({"event_ancestors", :ancestor_id})

  



  
    
      
    
    
      link_up(events_table, arg)



        
          
        

    

  


  

Create a link table for many-to-many event relationships.
Use this to add link tables in a separate migration after the events table exists.
Parameters
	events_table - The events table this link table references
	link_spec - Either {link_table, foreign_key} or {link_table, foreign_key, opts}

Options
	:typed - Add a type integer column to distinguish different relationship types
on the same link table (default: false)
	:constrained - Create a database trigger to enforce that linked events share the
same parent_id (default: true). Set to false for non-PostgreSQL databases.

Examples
Basic link table:
Spector.Migration.link_up("events", {"event_ancestors", :ancestor_id})
Link table with type column:
Spector.Migration.link_up("events", {"event_links", :linked_id, typed: true})
Link table without parent_id constraint (for non-PostgreSQL databases):
Spector.Migration.link_up("events", {"event_links", :linked_id, constrained: false})

  



  
    
      
    
    
      up(opts)



        
          
        

    

  


  

Create the event table(s) and any link tables.
Use :table for a single table or :shards for sharded setups.
See module documentation for available options.

  


        

      


  

    
Spector.Integrity.HashMismatch exception
    



      
Exception raised when hash chain verification fails.
This indicates that an event's hash does not match the expected hash computed
from the previous event's hash and the current event's data. This could indicate
data tampering or corruption.

      




  

    
Spector.Integrity.SavepointFailure exception
    



      
Exception raised when savepoint verification fails.
This indicates that replaying events through different paths (skipping vs applying
savepoints) produces divergent states, which means the savepoint/2 callback
is not correctly capturing all relevant state.

